
CUESTIÓN  1 (5 puntos) 

Se desea implementar un velocímetro para una motocicleta, empleando para ello un sensor (que generará 
un pulso de duración 1 ciclo de reloj cada vez que la rueda delantera de la motocicleta realice una vuelta 
completa) y una FPGA. En la FPGA se implementará un circuito que recibiendo los pulsos del sensor 
activará una tira de 8 LEDs, de manera proporcional a la velocidad de la motocicleta. La frecuencia de la 

señal de reloj que recibe la FPGA son 50 MHz. Se pide:  

a) Describir un módulo VHDL que a partir de la salida del sensor, el reloj de la FPGA y una señal de 
reset, genere una señal de 6 bits con el número de vueltas (codificado en binario) que ha dado la 
rueda de la motocicleta en el último segundo. Esta salida deberá actualizarse al final de cada 
segundo, manteniéndose estable durante todo el segundo siguiente. El máximo número de vueltas 
que se considerará es 63. 

b) Usando la descripción del apartado a como un componente jerárquico, describa la lógica que 
controla la activación de la tira de LEDs, de tal manera que el primer LED se activará cuando la 
velocidad proporcionada por el bloque anterior esté por debajo de 8 vueltas por segundo, el 
segundo se activará cuando la entrada entré entre 8 y 15 vueltas, y así sucesivamente hasta los 8 
LEDs que componen la tira.  

CUESTIÓN  2  (2 puntos) 

Describir en VHDL (entidad y arquitectura) la siguiente máquina de estados: 

 

CUESTIÓN  3  (3 puntos) 

Describir un módulo VHDL que se corresponda con un circuito que genere la secuencia síncrona 11011001 
de forma cíclica, de estas tres maneras posibles: 

c) Usando como base un contador binario síncrono 

d) Usando como base un contador Johnson 

e) Usando como base un contador en anillo 

 
 

Duración del examen: 1 hora y 30 minutos. 

1   library IEEE;
2   use IEEE.STD_LOGIC_1164.ALL;
3   use IEEE.NUMERIC_STD.ALL;
4   
5   entity velocimetro is
6   Port ( clk : in STD_LOGIC;
7   reset : in STD_LOGIC;
8   sensor : in STD_LOGIC;
9   VueltasOut : out std_logic_vector(5 downto 0));

10   end velocimetro;
11   
12   architecture Behavioral of velocimetro is
13   
14   constant Conta1segMAX : integer := 50*(10**6) - 1;
15   signal conta1seg : integer range 0 to Conta1segMAX;
16   signal UnSeg : std_logic;
17   
18   signal Vueltas: unsigned(5 downto 0);
19   
20   begin
21   
22   -- Divisor de Frecuencia a 1 segundo
23   DivFreq: process(clk,reset)
24   begin
25   if reset = '1' then
26   conta1seg <= 0;
27   elsif clk'event and clk = '1' then
28   if conta1seg < Conta1segMAX then
29   conta1seg <= conta1seg + 1;
30   else
31   conta1seg <= 0;
32   end if;
33   end if;
34   end process;
35   UnSeg <= '1' when conta1seg = Conta1segMAX else '0';
36   
37   -- Proceso en el que se cuentan las vueltas y se actualiza la salida (1 vez 

por segundo)
38   ContaVueltas: process(clk,reset)
39   begin
40   if reset = '1' then
41   Vueltas <= "000000";
42   VueltasOut <= "000000";
43   elsif clk'event and clk = '1' then
44   if Unseg = '1' then
45   VueltasOut <= std_logic_vector(Vueltas);
46   Vueltas <= "000000";
47   elsif sensor = '1' then
48   Vueltas <= Vueltas + 1;
49   end if;
50   end if;
51   end process;
52   
53   end Behavioral;
54   

1   library IEEE;
2   use IEEE.STD_LOGIC_1164.ALL;
3   use IEEE.NUMERIC_STD.ALL;
4   
5   entity TopVelocimetro is
6   Port ( clk : in STD_LOGIC;
7   reset : in STD_LOGIC;
8   sensor : in STD_LOGIC;
9   leds : out STD_LOGIC_VECTOR(7 DOWNTO 0)

10   );
11   end TopVelocimetro;
12   
13   architecture Behavioral of TopVelocimetro is
14   
15   component velocimetro is
16   Port ( clk : in STD_LOGIC;
17   reset : in STD_LOGIC;
18   sensor : in STD_LOGIC;
19   VueltasOut : out std_logic_vector(5 downto 0));
20   end component;
21   
22   signal VueltasOut : std_logic_vector(5 downto 0);
23   signal uV : unsigned(5 downto 0);
24   
25   begin
26   
27   -- Instancia del componente velocímetro
28   instVelo: velocimetro
29   port map(
30   clk => clk,
31   reset => reset,
32   sensor => sensor,
33   VueltasOut => VueltasOut
34   );
35   
36   --Transformación a unsigned para facilitar la codificación
37   uV <= unsigned(VueltasOut);
38   
39   leds <= "00000001" when uV < 8 else
40   "00000010" when uV < 16 else
41   "00000100" when uV < 24 else
42   "00001000" when uV < 32 else
43   "00010000" when uV < 40 else
44   "00100000" when uV < 48 else
45   "01000000" when uV < 56 else
46   "10000000";
47   end Behavioral;
48   

1   -------------------------
2   -- Electrónica Digital --
3   -- PEC diciembre 2017  --
4   -- Cuestión 3          --
5   -------------------------
6   
7   library ieee;
8   use ieee.std_logic_1164.all;
9   use ieee.numeric_std.all;

10   
11   entity c3 is
12   port (
13   clk : in std_logic;
14   reset : in std_logic;
15   seq : out std_logic
16   );
17   end c3;
18   
19   -------------------------------------------
20   -- La solución comienza a partir de aquí --
21   -------------------------------------------
22   
23   -- a) Base: contador binario síncrono
24   architecture cnt_binario of c3 is
25   signal cnt : unsigned(2 downto 0);
26   begin
27   
28   process(clk,reset)
29   begin
30   if reset = '1' then
31   cnt <= (others => '0');
32   elsif clk'event and clk = '1' then
33   cnt <= cnt + 1;
34   end if;
35   end process;
36   
37   with cnt select
38   seq <= '1' when "000",
39                  '1' when "001",
40                  '0' when "010",
41                  '1' when "011",
42                  '1' when "100",
43                  '0' when "101",
44                  '0' when "110",
45                  '1' when "111",
46                  '-' when others;
47   
48   end cnt_binario;
49   
50   -- b) Base: contador Johnson
51   architecture cnt_johnson of c3 is
52   signal shiftreg : std_logic_vector(3 downto 0);
53   begin
54   
55   process(clk,reset)
56   begin
57   if reset = '1' then
58   shiftreg <= (others => '0');
59   elsif clk'event and clk = '1' then
60   shiftreg <= shiftreg(2 downto 0) & not shiftreg(3);
61   end if;
62   end process;
63   
64   with shiftreg select
65   seq <= '1' when "0000",
66                  '1' when "0001",
67                  '0' when "0011",
68                  '1' when "0111",
69                  '1' when "1111",
70                  '0' when "1110",
71                  '0' when "1100",
72                  '1' when "1000",
73                  '-' when others;



74   
75   end cnt_johnson;
76   
77   -- c) Base: contador en anillo
78   architecture cnt_anillo of c3 is
79   signal shiftreg : std_logic_vector(7 downto 0);
80   begin
81   
82   process(clk,reset)
83   begin
84   if reset = '1' then
85   shiftreg <= (0 => '1', others => '0');
86   elsif clk'event and clk = '1' then
87   shiftreg <= shiftreg(6 downto 0) & shiftreg(7);
88   end if;
89   end process;
90   
91   with shiftreg select
92   seq <= '1' when "00000001",
93                  '1' when "00000010",
94                  '0' when "00000100",
95                  '1' when "00001000",
96                  '1' when "00010000",
97                  '0' when "00100000",
98                  '0' when "01000000",
99                  '1' when "10000000",

100                  '-' when others;
101   
102   end cnt_anillo;
103   

